
Lab 4: Proc Mixed: Fixed Effects

Overview

1. Construction of the pain data set.  

a. Time varying tmt effect

b. Log2 pain tolerance response

2. Class variables

a. cs has two levels

b. Estimate statements

c. tmt has four levels

d. tmt interpretation is awkward.  Redefine level names to improve interpretation.  

e. Estimate statements for various effects.  There are a lot!   

f. cs*tmt interaction

3. Alternative parameterizations of the model.  

a. Drop tmt effect, keep interaction.  

b. Drop cs effect, keep interaction.  

c. Include noint keyword in model statement.  

4. Contrast statements and more estimate statements.  

5. Continuous variables

a. Interactions 

b. Estimate statements

c. Contrast statements

6. Lsmeans statement

The Pediatric Pain data has up to 4 observations per subject.  The main response is the length of time in seconds that subjects can keep their hand immersed in cold water.  Subjects are classified into attenders or distracters (coping style or CS).  Before the 4th trial, subjects are given a counseling intervention to either attend, distract, or no advice.  The (time-varying) treatment variable has 4 levels, pre-treatment (zcontrol) and post-treatment: either attend, distract or none.  
Lab Tasks

1. Read in the pediatric pain data and build the data set.  We have done this before.  This also creates a time-varying ‘tmt’ variable and takes the log base 2 of pain tolerance for fitting purposes.  

data pain;

infile 'C:\Users\Rob\Courses\rm\SAShandouts\lab1\pain.txt';

*infile 'f:\class\m236\pain.txt'; /*CHANGE THIS PATH IF PAIN.TXT NOT IN THIS DIRECTORY*/
*WE WON'T BE USING OR RETAINING IN THE DATASET VAR3, VAR8-VAR9, VAR12-VAR13, VAR16-VAR19;
input id ses var3 age_months gendera treat coping_s var8-var9 d1 d2 var12-var13 d3 d4 var16-var19;

  format cs $10. treatment $13.;
*CHANGE MISSING CODE TO MISSING VALUES;
  array v(6) age_months ses d1 d2 d3 d4;

  do i=1 to 6;


if v(i)=-999 then v(i)=.;


end;

*CREATE COPING STYLE VARIABLE;  

  if coping_s=1 then cs='attender';

    else cs='distracter';

*THE VARIABLE "TREAT" IS RENAMED "TREATMENT" AS FOLLOWS;
  if treat=1 then treatment='attend';

    else if treat=2 then treatment='distract';

    else if treat=3 then treatment='no directions';

  if gendera=1 then gender='male';

    else if gendera=2 then gender='female';

  age = age_months/12;

  label age='age in years'
        cs='coping style'
        ses='socioeconomic status';

  run;

 * Convert the pain data from wide to long format.;
 * Create log response and time varying tmt variable.;
data painlong(keep = id ses treatment age gender cs paintol trial tmt l2paintol);

set pain;

*RESTRUCTURE DATASET TO LONG FORMAT WHICH IS THE FORMAT REQUIRED FOR PROC MIXED;
*D1, D2, D3, AND D4 ARE THE FOUR REPEATED MEASUREMENTS ON EACH SUBJECT;
  array d(4) d1 d2 d3 d4 ;

  do i=1 to 4;

    trial=i;

    paintol=d(i); 


l2paintol=log2(paintol);

    tmt = 'baseline';  
*This will get us in modest difficulty. We’ll come back and fix this with the next statement;
*    tmt = 'zbaseline';

    if i = 4 then tmt = treatment;

    output;

    end;

label 
paintol='Pain Tolerance'


tmt = 'Time Varying Treatment';

run;

* Sort by id and trial.  ;
proc sort data=painlong; by id trial;

run;

* Check the data came out ok; 

proc contents data=painlong;

run;

proc print data=painlong;

run;

OPTIONS CENTER LINESIZE=81 PAGESIZE=66;

2. Class variables

a. Use cs as a class variable to predict log base 2 pain tolerance.  Give each model a separate title.  

TITLE1 'cs';

proc mixed method=ml data=painlong covtest;

class id cs;

model l2paintol = cs / s;

random intercept / sub=id type=un;

run;

Keywords and commands:  

proc mixed – Telling SAS to use the mixed procedure…

method=ml – use Maximum Likelihood instead of REML

covtest – standard errors and t-tests for covariance parameters

class – identifying the categorical variables 

model – the response = <list of predictors>

/ s – parameter estimates for the fixed effects.

random – this is a random intercept random effects model.  

/ sub=id telling Proc Mixed how to identify observations from the same subject

/ type=un – the covariance matrix “D” of the random effects.  

Find the following sections of the output:  

                               Model Information

                             Class Level Information

                                   Dimensions

                                Iteration History

                           Convergence criteria met
                         Covariance Parameter Estimates

                                 Fit Statistics

                        Null Model Likelihood Ratio Test

                          Solution for Fixed Effects

                       Type 3 Tests of Fixed Effects

Also look in the log to check on convergence.  This is important to do each time.
b. Estimate statements for estimating linear combinations of parameters.  They go after the random (or repeated) statement.  

The cl keyword causes Mixed to print a confidence interval for that particular estimate.  The three estimate statements produce an estimate of the attender and distracter means and an estimate of the difference between distracters and attenders.  

estimate 'A' intercept 1 cs 1 0 ;

estimate 'D' intercept 1 cs 0 1 ;

estimate 'D-A' cs -1 1 / cl;

Include those three estimate statements in the Proc Mixed code.  Since the second level of the cs effect has been forced equal to zero, you would think that 

estimate 'D-A' cs -1 0 / cl;  *WRONG!;
would give the same results.  Try it and see.  

c. Add the tmt main effect to the model.  

i. Tmt goes after cs in the model statement.  

ii. Treatment must go in the class statement as well.  

iii. Take out the estimate statements.  

iv. Change the title!  

class id cs tmt;

model l2paintol = cs tmt / s;

There is one change to one item in the ‘dimension’ section and the solution for fixed effects and the type 3 test sections have larger changes.  

TODO:  Identify the changes in these sections.  

d. The interpretation of tmt is slightly awkward.  SAS sets the alphabetically last level of the class variable to zero.  Better is to have baseline as the forced to zero level.   

Redefine baseline as 'zbaseline' back in the original code where we switch from wide to long format.  Change:   tmt = 'baseline'; To     tmt = 'zbaseline';  Now rerun the code creating the painlong data set.  

Rerun the proc mixed program.  

TODO Now attend distract and no_directions are estimated mean changes (on log scale) due to the treatment effects.  

TODO Locate output of the F test for no tmt effect.  F=3.11, on 3,178 df.  Use this.  
TODO Can also do likelihood ratio test (LRT) for H0 no treatment effect.  Subtract -2 log likelihood for this model, and model with cs only in it.  587.3 - 578.1  = 9.2 ~ chisquare (3).  Question: What is the relationship between 2=9.2 and F=3.11?  
(WARNINGS: 1. Previous F-test is easier to use; I often use the F-test, rarely the LRT.  2.  You can get a LRT for the fixed effects ONLY if you have keyword METHOD=ML in the Proc Mixed statement.  ) 
e. Estimate statements for treatment effects and the baseline effects.  

i. Include estimate statements in the previous proc mixed program to calculate:  

1. Statements 1, 2 estimate baseline means.

2. Statements 3, 4 attempt to estimate D baseline.  Neither is quite right, both give the same answer.   

3. Statement 5 is the baseline difference.

4. Statements 6-8 are the means of attenders after the 3 treatments.  

ii. TODO: You should create 

1. 9-11 calculate the means of distracters after the 3 treatments.  

2. 12-14 there are 3 pairwise differences between treatment effects, calculate those as well.   Here is one:  

estimate 'tmt A-N' tmt 1 0 -1 0;

* Need estimate statements to estimate means before and after treatment.  ;
* estimate statements go just before the run statement;

* These are the first 8 estimate statements, you create 6 more.  ;
estimate '1 A baseline' intercept 1 cs 1 0 tmt 0 0 0 1;

estimate '2 D baseline' intercept 1 cs 0 1 tmt 0 0 0 1;

estimate '3 D baseline alt' intercept 1 cs 0 1 ; *compare to previous;
estimate '4 D baseline alt2' intercept 1 cs 0 1 tmt .25 .25 .25 .25; 
estimate '5 D-A' cs -1 1 ;

estimate '6 A A' intercept 1 cs 1 0 tmt 1 0 0 0;

estimate '7 A D' intercept 1 cs 1 0 tmt 0 1 0 0;

estimate '8 A N' intercept 1 cs 1 0 tmt 0 0 1 0;

f. Now add the interaction cs*tmt into the mean model.  

TITLE1 'cs tmt cs*tmt';

proc mixed method=ml data=painlong covtest;

class id cs tmt;

model l2paintol = cs tmt cs*tmt / s;

random intercept / sub=id type=un;

run;

Interpretation of cs*tmt effect is differences for each treatment for cs=attenders versus cs=distracters.  The tmt effects are the effects of treatment on distracters.  Without estimate statements it will be annoying to figure out exact effects of treatment on attenders.  
3. Some alternative parameterizations can be gotten from SAS.  
a. Remove tmt from model statement, keep interaction; Change title!  
TITLE1 'cs tmt cs*tmt alt1';

proc mixed method=ml data=painlong covtest;

class id cs tmt;

model l2paintol = cs cs*tmt / s;

random intercept / sub=id type=un;

run;

Now cs*tmt are the 6 treatment effects, at trial 4 minus baseline for both distracters and attenders.  This is one of my preferred parameterizations.  
b. Delete cs from the model statement.  
TITLE1 'cs tmt cs*tmt alt2';

proc mixed method=ml data=painlong covtest;

class id cs tmt;

model l2paintol = cs*tmt / s;

random intercept / sub=id type=un;

run;

The intercept is distract mean at baseline.  All other effects are estimates of differences between various means (AA AD AN AB DA DD DN) and distracter baseline (DB).  
c. Include the no intercept keyword noint.
TITLE1 'cs tmt cs*tmt alt3';

proc mixed method=ml data=painlong covtest;

class id cs tmt;

model l2paintol = cs*tmt / s noint;

random intercept / sub=id type=un;

run;

The effects are now the 8 group means at the 8 combinations of cs*tmt.  No matter which way we fit the model, a large number of estimate statements will be useful for estimating all combinations of contrasts.  I like this parameterization too.  
4. Contrast statements and many Estimate statements are usually necessary.  

There are many estimate statements to identify everything that we want to estimate in this model.  If there are not 8 numbers after the cs*tmt statement, then SAS pads out the list with zeros.  We may want to test whether a subset of the covariates = 0, this requires the contrast statement.  

TITLE1 'cs*tmt + all estimates of interest';

proc mixed method=reml covtest cl;

class id cs tmt;

model l2paintol=cs tmt cs*tmt /  s cl;

repeated / type=cs subject=id;

* 8 means;
estimate 'AB' intercept 1 cs 1 0 tmt 0 0 0 1 cs*tmt 0 0 0 1  ;

estimate 'DB' intercept 1 cs 0 1 tmt 0 0 0 1 cs*tmt 0 0 0 0 0 0 0 1  ;

estimate 'AA' intercept 1 cs 1 0 tmt 1 0 0 0 cs*tmt 1  ;

estimate 'AD' intercept 1 cs 1 0 tmt 0 1 0 0 cs*tmt 0 1  ;

estimate 'AN' intercept 1 cs 1 0 tmt 0 0 1 0 cs*tmt 0 0 1  ;

estimate 'DA' intercept 1 cs 0 1 tmt 1 0 0 0 cs*tmt 0 0 0 0 1  ;

estimate 'DD' intercept 1 cs 0 1 tmt 0 1 0 0 cs*tmt 0 0 0 0 0 1  ;

estimate 'DN' intercept 1 cs 0 1 tmt 0 0 1 0 cs*tmt 0 0 0 0 0 0 1  ;

* all 3 pairwise comparisons among 3 tmt effects for attenders;
estimate 'tmt D - A | A' tmt -1 1 0 0 cs*tmt -1 1 0 0 0 0 0 0  ;

estimate 'tmt D - N | A' tmt 0 1 -1 0 cs*tmt 0 1 -1 0 0 0 0 0  ;

estimate 'tmt A - N | A' tmt 1 0 -1 0 cs*tmt 1 0 -1 0 0 0 0 0  ;

*three treatment effects for attenders;
estimate 'tmt A - B | A' tmt 1 0 0 -1 cs*tmt 1 0 0 -1 0 0 0 0  ;

estimate 'tmt D - B | A' tmt 0 1 0 -1 cs*tmt 0 1 0 -1 0 0 0 0  ;

estimate 'tmt N - B | A' tmt 0 0 1 -1 cs*tmt 0 0 1 -1 0 0 0 0  ;

* all 3 pairwise comparisons among 3 tmt effects for distracters;
estimate 'tmt D - A | D' tmt -1 1 0 0 cs*tmt 0 0 0 0 -1 1 0 0  ;

estimate 'tmt D - N | D' tmt 0 1 -1 0 cs*tmt 0 0 0 0 0 1 -1 0  ;

estimate 'tmt A - N | D' tmt 1 0 -1 0 cs*tmt 0 0 0 0 1 0 -1 0  ;

*three treatment effects for distracters;
estimate 'tmt A - B | D' tmt 1 0 0 -1 cs*tmt 0 0 0 0 1 0 0 -1  ;

estimate 'tmt D - B | D' tmt 0 1 0 -1 cs*tmt 0 0 0 0 0 1 0 -1  ;

estimate 'tmt N - B | D' tmt 0 0 1 -1 cs*tmt 0 0 0 0 0 0 1 -1  ;

contrast 'H0: no tmt effect for distracters' tmt 1 0 0 -1 cs*tmt 0 0 0 0 1 0 0 -1  ,

 tmt 0 1 0 -1 cs*tmt 0 0 0 0 0 1 0 -1  , tmt 0 0 1 -1 cs*tmt 0 0 0 0 0 0 1 -1  ;

contrast 'H0: all tmt effects equal for distracters' tmt -1 1 0 0 cs*tmt 0 0 0 0 -1 1 0 0  , tmt 0 1 -1 0 cs*tmt 0 0 0 0 0 1 -1 0 , tmt 1 0 -1 0 cs*tmt 0 0 0 0 1 0 -1 0  ;

run;

The keyword cl in the proc mixed statement creates confidence interval for covariance parameters.  The keyword cl in the model statement creates confidence intervals for fixed effects.  

We usually use REML for the final model fit, when we have settled on the fixed effects.  
TODO:  Confirm in your mind what all the estimate statements do. 
TODO:  The two contrast statements test hypotheses about the tmts for the distracters.  Why does one have 2 df and one have 3 df, even though both appear to have 3 quantities equal to zero under H0?  
A. Create the corresponding contrast statements for the attenders.  Include them in the model and run it.  Compare results for distracters and for attenders.  

5. Continuous variables.

Include age, ses and gender in the analysis.  The estimate statement estimates the mean of a 9yr old female distracter with ses=100 taught to distract.  The contrast statements tests for whether all three age ses gender coefficients are zero.  

TITLE1 'cs tmt cs*tmt age ses gender';

proc mixed method=ml data=painlong covtest;

class id cs tmt gender;

model l2paintol = cs tmt cs*tmt age ses gender/ s;

random intercept / sub=id type=un;

estimate '9yr dd ses 100 gender f' intercept 1 cs 0 1 tmt 0 1 0 0 



cs*tmt 0 0 0 0 0 1 0 0 age 9 ses 100 gender 1 0;

contrast 'three vars' age 9, ses 70, gender 1 -1;

contrast 'three vars' age 1, ses 1, gender 1 -1;

lsmeans cs;

run;

Both contrast statements give the same result.  Explain why.  Change the contrast to have 'gender 1 0' and notice that nothing comes out in the results.  The LSMEANS statement estimates the different cs levels at the average values of the other statements.  I think it puts in .25 for the four tmt levels, but am not completely sure.  (How could you confirm this?)  

6. LSMEANS statements.

LSMEANS statements are good for getting a large number of estimates out when you have class variables.  You have to be careful about what the estimates mean from an lsmeans statement.  The first LSMEAN statement below produces estimates at the average value of age.  The second produces estimates at age=8 years old.  The third also produces all pairwise differences, which means there is a lot of output.  Not all comparisons may be of interest, but this does produce all the interesting contrasts that we got out using estimate statements earlier.  However, you have to plow through and figure out which ones you want and how to interpret them.  

TITLE1 'cs tmt cs*tmt age;

proc mixed method=ml data=painlong covtest;

class id cs tmt;

model l2paintol = cs tmt cs*tmt age / s;

random intercept / sub=id type=un;
*e specifies that we want the matrix – if we look down

 these columns we can see how to code each estimate;

lsmeans cs*tmt / e;

lsmeans cs*tmt / at age = 8 e;

lsmeans cs*tmt / e diff;

run;

